Demo 4: Discrete Systems with Random Elements

In this demo we will examine what happens when random elements are introduced into a second order discrete system. To see that, we will consider the following system of difference equations

$$x_1(k+1) = -0.5x_1(k) + a(k)x_2(k)$$
$$x_2(k+1) = b(k)x_1(k) - 0.7x_2(k)$$

where pair $(a(k), b(k))$ is *randomly* chosen in each step from $(-0.25, 0.75)$, $(0.5, 0.5)$ and $(0.75, -0.5)$. One way to make such a choice involves Scilab function

w=rand()

which produces a random number w between 0 and 1. If $w \leq 0.33$, we will set $a(k) = -0.25$ and $b(k) = 0.75$. If it turns out that $0.33 < w \leq 0.66$, $a(k)$ and $b(k)$ will be chosen as $a(k) = b(k) = 0.5$ and if $0.66 < w \leq 1$, these coefficients will take on values $a(k) = 0.75$ and $b(k) = -0.5$.

The file dem4.sci shown below (which describes the right hand side of our equation) makes use of this simple procedure. In this function, the input parameters q1, q2, and q3 correspond to vectors $q_1 = [-0.25; 0.75]$, $q_2 = [0.5; 0.5]$and $q_3 = [0.75; -0.5]$ from which the pair $(a(k), b(k))$ is chosen.

```
function y=dem4(t,x,q1,q2,q3)
  w=rand();
  if w<=0.33 then
     a=q1(1);
     b=q1(2);
  elseif (w>0.33)&(w<=0.66)
     a=q2(1);
     b=q2(2);
  else
     a=q3(1);
     b=q3(2);
  end
  y(1)=-0.5*x(1)+a*x(2);
  y(2)=b*x(1)-0.7*x(2);
endfunction
```

To solve this system with initial condition $x(0) = [1; 1]$ and $k_0 = 0$ for $k = 0, 1, 2, \ldots, 50$, you will need to type the following sequence of commands:

```
q1=[-0.25;0.75];
q2=[0.5;0.5];
q3=[0.75;-0.5];
x0=[1;1];
k0=0;
kvect=0:1:50;
y=ode("discrete",x0,k0,kvect,list(dem4,q1,q2,q3));
```

Since

$$y = \begin{bmatrix} x_1(0) & x_1(1) & x_1(2) & \ldots & x_1(49) & x_1(50) \\ x_2(0) & x_2(1) & x_2(2) & \ldots & x_2(49) & x_2(50) \end{bmatrix}$$

we can extract $x_1(k)$ and $x_2(k)$ using commands

```
x1=y(1,:);
```

and

```
x2=y(2,:);
```

respectively.

To plot the obtained sequences, enter

```
plot(kvect,x1,'.',kvect,x1)
```

and

```
plot(kvect,x2,'.',kvect,x2)
```

The resulting graphs are shown in Figs. 1 and 2.

Note that solving this equation again will *not* produce the same sequence, even if the initial conditions remain unchanged. This is due to the fact that the choice between q_1, q_2 and q_3 is made *randomly*, and is therefore different every time we solve the equation. Figs. 3 and 4 illustrate what I obtained after running this program for a second time (your results will undoubtedly be different).


Figure 1
X1(K)
K
1
0.8
0.6
0.4
0.2
0
-0.2
-0.4
-0.6
-0.8
0
5
10
15
20
25
30
35
40
45
50


Figure 2
X2(K)
K
1
0.9
0.8
0.7
0.6
0.5
0.4
0.3
0.2
0.1
0
-0.1
-0.2
-0.3
-0.4
0
5
10
15
20
25
30
35
40
45
50



Figure 3
1
0.9
0.8
0.7
0.6
0.5
0.4
0.3
0.2
0.1
0
-0.1
-0.2
X1(K)
0
5
10
15
20
25
30
35
40
45
50
K


Figure 4
X2(K)
K
1
0.9
0.8
0.7
0.6
0.5
0.4
0.3
0.2
0.1
0
-0.1
-0.2
0
5
10
15
20
25
30
35
40
45
50